# $Id: PKGBUILD 80790 2012-12-03 18:01:33Z ttoepper $
# Maintainer: Thorsten Töpper <atsutane-tu@freethoughts.de>
# Contributor: Jelle van der Waa <jelle vdwaa nl>
# Contributor: Thomas Holmquist <thomas@vorget.com>
# Contributor: Jan Boehringer <ja(a)bm-boehringer.de>
# Contributor: Alex Griffin <griffin.aj(a)gmail.com>

pkgname=polipo
pkgver=1.0.4.1
pkgrel=9
pkgdesc="A small and fast caching web proxy."
arch=('i686' 'x86_64')
url="http://www.pps.jussieu.fr/~jch/software/polipo/"
license=('GPL')
depends=('bash')
makedepends=('texinfo')
install=polipo.install
source=("http://freehaven.net/~chrisd/polipo/polipo-$pkgver.tar.gz"
        "polipo.rc.d" "polipo.conf.d" "polipo.service")
md5sums=('bfc5c85289519658280e093a270d6703'
         '7406aad26807fb3a4e5f7b01ecb97a59'
         '685aa0c6070dee11c701932d23afcc6a'
         '143ef45db99dc745b6c3a606847be9fe')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make 	PREFIX="$pkgdir/usr" \
        MANDIR="$pkgdir/usr/share/man" \
        INFODIR="$pkgdir/usr/share/info" \
        LOCAL_ROOT="$pkgdir/usr/share/polipo/www" \
        DISK_CACHE_ROOT="$pkgdir/var/cache/polipo" \
        install 


  # install config files
  install -Dm 644 config.sample $pkgdir/etc/polipo/config.sample
  install -Dm 644 forbidden.sample $pkgdir/etc/polipo/forbidden.sample

  # install license
  install -Dm644 COPYING $pkgdir/usr/share/licenses/polipo/LICENSE

  # install daemon scripts / cron
  install -Dm 755  $srcdir/polipo.rc.d $pkgdir/etc/rc.d/polipo
  install -Dm 644  $srcdir/polipo.service $pkgdir/usr/lib/systemd/system/polipo.service
  install -Dm 644  $srcdir/polipo.conf.d $pkgdir/etc/conf.d/polipo.conf
}

# vim:set ts=2 sw=2 et:
